

# NAG Library Routine Document

## F06WAF (DLANSF)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F06WAF (DLANSF) returns the value of the 1-norm, the  $\infty$ -norm, the Frobenius norm, or the maximum absolute value of the elements of a real symmetric matrix  $A$  stored in Rectangular Full Packed (RFP) format.

### 2 Specification

```
FUNCTION F06WAF (NORM, TRANSR, UPLO, N, AR, WORK)
REAL (KIND=nag_wp) F06WAF
INTEGER                N
REAL (KIND=nag_wp) AR(N*(N+1)/2), WORK(*)
CHARACTER(1)          NORM, TRANSR, UPLO
```

The routine may be called by its LAPACK name *dlansf*.

### 3 Description

Given a real  $n$  by  $n$  symmetric matrix,  $A$ , F06WAF (DLANSF) calculates one of the values given by

$$\|A\|_1 = \max_j \sum_{i=1}^n |a_{ij}| \quad (\text{the 1-norm of } A),$$

$$\|A\|_\infty = \max_i \sum_{j=1}^n |a_{ij}| \quad (\text{the } \infty\text{-norm of } A),$$

$$\|A\|_F = \left( \sum_{i=1}^n \sum_{j=1}^n |a_{ij}|^2 \right)^{1/2} \quad (\text{the Frobenius norm of } A), \quad \text{or}$$

$$\max_{i,j} |a_{ij}| \quad (\text{the maximum absolute element value of } A).$$

$A$  is stored in compact form using the RFP format. The RFP storage format is described in Section 3.3.3 in the F07 Chapter Introduction.

### 4 References

Basic Linear Algebra Subprograms Technical (BLAST) Forum (2001) *Basic Linear Algebra Subprograms Technical (BLAST) Forum Standard* University of Tennessee, Knoxville, Tennessee <http://www.netlib.org/blas/blast-forum/blas-report.pdf>

Gustavson F G, Waśniewski J, Dongarra J J and Langou J (2010) Rectangular full packed format for Cholesky's algorithm: factorization, solution, and inversion *ACM Trans. Math. Software* **37**, 2

## 5 Arguments

- 1: NORM – CHARACTER(1) *Input*
- On entry:* specifies the value to be returned.
- NORM = '1' or 'O'  
The 1-norm.
- NORM = 'I'  
The  $\infty$ -norm.
- NORM = 'F' or 'E'  
The Frobenius (or Euclidean) norm.
- NORM = 'M'  
The value  $\max_{i,j} |a_{ij}|$  (not a norm).
- Constraint:* NORM = '1', 'O', 'I', 'F', 'E' or 'M'.
- 2: TRANSR – CHARACTER(1) *Input*
- On entry:* specifies whether the RFP representation of  $A$  is normal or transposed.
- TRANSR = 'N'  
The matrix  $A$  is stored in normal RFP format.
- TRANSR = 'T'  
The matrix  $A$  is stored in transposed RFP format.
- Constraint:* TRANSR = 'N' or 'T'.
- 3: UPLO – CHARACTER(1) *Input*
- On entry:* specifies whether the upper or lower triangular part of  $A$  is stored.
- UPLO = 'U'  
The upper triangular part of  $A$  is stored.
- UPLO = 'L'  
The lower triangular part of  $A$  is stored.
- Constraint:* UPLO = 'U' or 'L'.
- 4: N – INTEGER *Input*
- On entry:*  $n$ , the order of the matrix  $A$ .
- When N = 0, F06WAF (DLANSF) returns zero.
- Constraint:*  $N \geq 0$ .
- 5: AR( $N \times (N + 1)/2$ ) – REAL (KIND=nag\_wp) array *Input*
- On entry:* the upper or lower triangular part (as specified by UPLO) of the  $n$  by  $n$  symmetric matrix  $A$ , in either normal or transposed RFP format (as specified by TRANSR). The storage format is described in detail in Section 3.3.3 in the F07 Chapter Introduction.
- 6: WORK(\*) – REAL (KIND=nag\_wp) array *Workspace*
- Note:** the dimension of the array WORK must be at least  $\max(1, N)$  if NORM = '1', 'O' or 'I', and at least 1 otherwise.

## 6 Error Indicators and Warnings

None.

## 7 Accuracy

The BLAS standard requires accurate implementations which avoid unnecessary over/underflow (see Section 2.7 of Basic Linear Algebra Subprograms Technical (BLAST) Forum (2001)).

## 8 Parallelism and Performance

F06WAF (DLANSF) is not threaded in any implementation.

## 9 Further Comments

None.

## 10 Example

This example reads in the lower triangular part of a symmetric matrix, converts this to RFP format, then calculates the norm of the matrix for each of the available norm types.

### 10.1 Program Text

```

Program f06wafe

!      F06WAF Example Program Text
!
!      Mark 26 Release. NAG Copyright 2016.
!
!      .. Use Statements ..
Use nag_library, Only: dlansf, dtrttf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: r_fro, r_inf, r_max, r_one
Integer                    :: i, info, lda, n
Character (1)              :: transr, uplo
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:,,:), ar(:), work(:)
!      .. Executable Statements ..
Write (nout,*) 'F06WAF Example Program Results'

!      Skip heading in data file
Read (nin,*)

      Read (nin,*) n, uplo, transr

      lda = n
      Allocate (a(lda,n),ar((n*(n+1))/2),work(n))

!      Read upper or lower triangle of matrix A from data file

      If (uplo=='L' .Or. uplo=='l') Then
        Do i = 1, n
          Read (nin,*) a(i,1:i)
        End Do
      Else
        Do i = 1, n
          Read (nin,*) a(i,i:n)
        End Do
      End If

!      Convert A to rectangular full packed storage in ar

!      The NAG name equivalent of dtrttf is f0lvef
      Call dtrttf(transr,uplo,n,a,lda,ar,info)

```

```

Write (nout,*)
Write (nout,99999)
'Norms of symmetric matrix stored in RFP format in ar:'
Write (nout,*)

! The NAG name equivalent of dlansf is f06waf
r_one = dlansf('1-norm',transr,uplo,n,ar,work)
Write (nout,99998) 'One norm          = ', r_one

r_inf = dlansf('Infinity',transr,uplo,n,ar,work)
Write (nout,99998) 'Infinity norm     = ', r_inf

r_fro = dlansf('Frobenius',transr,uplo,n,ar,work)
Write (nout,99998) 'Frobenius norm    = ', r_fro

r_max = dlansf('Max norm',transr,uplo,n,ar,work)
Write (nout,99998) 'Maximum norm     = ', r_max

99999 Format (1X,A)
99998 Format (1X,A,F9.4)
End Program f06waf

```

## 10.2 Program Data

```

F06WAF Example Program Data
6 'L' 'N' : N, UPLO, TRANSR
1.0
2.0 2.0
3.0 3.0 3.0
4.0 4.0 4.0 4.0
5.0 5.0 5.0 5.0 5.0
6.0 6.0 6.0 6.0 6.0 6.0 : Matrix A

```

## 10.3 Program Results

F06WAF Example Program Results

Norms of symmetric matrix stored in RFP format in ar:

```

One norm          = 36.0000
Infinity norm     = 36.0000
Frobenius norm    = 28.1247
Maximum norm     = 6.0000

```

---